Aquacultural Engineering 23 (2000) 95–102
aquacultural engineering
www.elsevier.nl/locate/aqua-online

Using computer spreadsheets for water flow and biofilter sizing in recirculating aquaculture production systems

Thomas M. Losordo [a,*], Alexander O. Hobbs [b]

[a] *Departments of Zoology and Biological and Agricultural Engineering, North Carolina State University, Campus Box 7646, Raleigh, NC 27695, USA*
[b] *Carolina Power and Light Company, P.O. Box 1551, Raleigh, NC, USA*

Received 16 December 1998; accepted 3 March 1999

Abstract

North Carolina State University has been active in the development, evaluation and demonstration of recirculating aquaculture technology since 1989. In the process, numerous engineering and economic spreadsheets (worksheets) have been developed to assist in the design and analysis of these systems. The spreadsheet described in this paper is based on a set of mass balance equations developed and described by Losordo and Westers. Aquaculture Water Reuse Systems: Engineering Design and Management. Developments in Aquaculture and Fisheries Sciences vol. 27, 1994, pp. 9–60) to estimate the carrying capacity and required flow rates of recirculating aquaculture production systems. This spreadsheet can be used to estimate the recycle flowrate that is required to maintain user specified water quality conditions for a given feed input rate and water treatment system configuration. These water quality conditions include suspended solids, total ammonia-nitrogen, and dissolved oxygen concentration. The spreadsheet also provides an estimate of the new water required by the system to maintain a user specified nitrate-nitrogen concentration. In addition, the spreadsheet provides an estimate of the required biofilter volume and cross-sectional (top) surface area for the given biofilter shape, depth and specific surface area of the biofilter media. The mass balance equations used in this spreadsheet are based upon waste metabolites generated and oxygen consumed by daily inputs of feed into a system. © 2000 Elsevier Science B.V. All rights reserved.

Keywords: Recirculating system; Spreadsheet model; Systems design

* Corresponding author. Tel.: + 1-919-515-7587; fax: + 1-919-515-5110.
E-mail address: tlosordo@unity.ncsu.edu (T.M. Losordo)

PII: S0144-8609(00)00048-0

1. Introduction

Interest in tank based aquaculture production systems using water recirculation is growing world-wide. Limitations of existing water supplies, the desire for increased system carrying capacity, reduction of heat loss, and reduction of waste effluent stream volumes are a few of the reasons for the interest in recirculating aquaculture technology. In the past, many recirculating production systems have been developed by designers with less than complete knowledge of basic engineering principles using a trial and error approach or past experience of what did and did not work. The accurate estimation of recycled water flow rate requirements and appropriate sizing of biological filters is of critical importance to the successful design of any tank-based system using recirculation technology. A design approach based on mass balance analysis was outlined by Losordo and Westers (1994). Using mass balance analysis a designer can estimate the flow requirements from the culture tank to water treatment components in order to control the build-up of waste solids, ammonia-nitrogen, and nitrate-nitrogen, and estimate the required flow to maintain an adequate dissolved oxygen concentration.

The input data used in the example spreadsheet in this paper are appropriate for the growout systems and conditions at the Carolina Power and Light Company (CP&L) Fish Barn project as described by Hobbs et al. (1997). Some of the input data describing the solids removal characteristics and nitrification rates of the system are from previous studies on a smaller, yet similar system at the North Carolina Fish Barn that has been described by Twarowska et al. (1997). The system being modeled utilizes round tanks (e.g. completely mixed conditions), a particle trap on the bottom (center) of the tank, a drum screen filter, a trickling biological filter, and down-flow bubble contactor for oxygenation. The studies by Twarowska et al. (1997) have shown that the particle trap and drum screen filter remove equal amounts (approximately 50% each) of wasted solids generated by tilapia. The actual flowrate through the CP&L Fish Barn water treatment system is approximately 795 l per minute (210 gpm) for each 60 m^3 culture tank.

2. Spreadsheet review and description

The spreadsheet as it appears on a computer can be viewed in Fig. 1. This spreadsheet was executed in Microsoft Excel[1], however it can be converted and used in any popular spreadsheet format (Claris Works, Lotus 123, etc.). The spreadsheet equations are also shown in a 'display formula' format in Column D of Fig. 1 so that it can be recreated by the reader and modified for specific uses.

The calculations in this spreadsheet are based upon information and data that are input by the user into the spreadsheet. Some of this information requires knowledge gained from previous experience with a similar system, components, fish

[1] Mention of a product or tradename does not constitute an endorsement of the product to the exclusion of other products by North Carolina State University or Carolina Power and Light Company.

	A	B	C	D
1	Spreadsheet for Flow Rate Estimation and Biofilter Sizing			
2	Copyright by NC State University 1998			
3				
4	*2.1. Tank Size and Biomass*	Values	Units	Calculation Formula
5				
6	Tank water depth	2.00	m	2.00
7	Tank radius	3.10	m	3.10
8	Tank volume	60	m^3	=B6*(PI()*(B7*B7))
9	Maximum culture density	60	kg/m^3	60
10	Fish biomass	3623	kgs	=B8*B9
11	Fish count	6000		6000
12	Fish weight	603.8	gm	=1000*B10/B11
13	Feed rate as % of body weight	1.25%		0.0125
14	Feed rate	45.3	kg/day	=B10*B13
15				
16	*2.2. TAN Mass Balance Calculations*			
17				
18	Feed protein content	38%		0.38
19	Total Ammonia Nitrogen (TAN) production rate	1.119	kg/day	=0.065*B14*B18
20	% TAN from feed	2.47%		=B19/B14
21	Desired TAN concentration in recirc water	1.8	mg/L	1.8
22	Passive nitrification	10%		0.10
23	TAN available after passive nitrification	1.007	kg/day	=B19*(1-B22)
24	Passive denitrification	0%		0.00
25	Maximum nitrate concentration desired	150	mg/L	150
26	New water required maintain nitrate concentration	6711	L/day	=((B23*1000000*(1-B24))/B25)
27	TAN available to Biofilter after effluent removal	0.995	kg/day	=B23-(B21/1000000)*B26
28	Biofilter efficiency for TAN removal	50%		0.50
29	Flow rate to remove TAN to desired concentration	1105143	L/day	=(B27)/(B28*(B21/1000000))
30		767	L/min	=B30/1440
31		203	gal/min	=B31/3.785
32				
33	*2.3. Biofilter Sizing Calculation*			
34				
35	Estimated nitrification rate	0.45	g TAN/m^2/day	0.45
36	Active nitrification surface required at rate	2210	m^2	=(B27)/(B35/1000)
37	Surface area of media	200	m^2/m^3	200
38	Total volume media	11.05	m^3	=B36/B37
39	Media unit price	$200	$/m^3	200
40	Media Cost	$2,210.29	media cost	=B39*B38
41	Media depth	1.65	m	1.65
42	Volume / depth yields face area	6.70	m^2	=B38/B41
43	Diameter of biofilter	2.92	m	=2*SQRT(B42/3.1416)
44				
45	*2.4. Solids Mass Balance Calculations*			
46				
47	Estimated percentage of feed becoming solid waste	25%		0.25
48	Waste solids produced	11.32	kg/day	=B14*B47
49	Desired SS conc.	10	mg/L	10
50	Est. % removed by particle trap	50%	particle trap	0.50
51	Waste solids remaining after particle trap	5.66	kg/day	=B48*(1-B50)
52	Waste solids remaining solids removal in effluent	5.59	kg/day	=(B51-(B49*B26/1000000))
53	Settling tank, bead filter, drum filter, etc. efficiency	50%		0.50
54	Flow rate to remove SS to desired concentration	1118729	L/day	=(B52)/(B53*(B49/1000000))
55		777	L/min	=B55/1440
56		205	gal/min	=B56/3.785
57				
58				
59	*2.5. Oxygen Mass Balance Calculations*			
60				
61	Submerged filter? (1=yes and 0=no)	0		0
62	Oxygen used / kg Feed	30%		0.30
63	Oxygen used by feed addition	13.59	kg/day	=B14*B62
64	Desired oxygen concentration in tank	5.0	mg/L	5.0
65	Dissolved oxygen concentration supplied to tank	18.0	mg/L	18.0
66	Oxygen used by passive nitrification	0.51	kg/day	=(B19-B23)*4.57
67	Oxygen used for nitrification in biofilter	0.00	kg/day	=B61*(B27)*4.57
68	Total oxygen used	14.10	kg/day	=B63+B66+B67
69	Estimated flow rate	1084385	L/day	=B68/((B65-B64)/1000000)
70		753	L/min	=B69/1440
71		199.0	gpm	=B70/3.785

Fig. 1. Spreadsheet for flowrate estimation and biofilter sizing.

species or feed type. Required input data are located in Column B of the spreadsheet and are offset in **bold** type. These input data cells are also offset in the text of this paper with **bold** type. Various areas of the spreadsheet are grouped and have headings displayed in *italics*.

2.1. Tank size and biomass calculations

Water quality within a recirculating aquaculture system is, to a large degree, a function of the tank size, the biomass in the tank, the rate of feed input, and the waste removal and treatment efficiency of the system. To start using the spreadsheet, the user must input some tank dimensions and the stocking rate information. In the example spreadsheet the tank depth of 2 m is input in cell **B6** while the tank radius of 3.1 m is input in cell **B7.** Tank volume is calculated and displayed in cell B8. The user then specifies the maximum culture density in cell **B9** (60 kg/m^3 in our example). The user then inputs the population of fish in cell **B11** which yields market size fish as calculated in cell B12. The feed-rate as % of body weight estimate is species and size specific and can be found in the literature or can be obtained from feed suppliers who have experience with the fish to be cultured. It should be noted that this spreadsheet was developed with the assumption that daily inputs of feed are spread over a 24-h period. The waste treatment capacity is affected if the feed is input over a shorter period of time because the spreadsheet will underestimate the water flow requirements at 'peak' periods of feed input. The daily feed rate is calculated and displayed in cell B14 and is used in the all of the calculations that follow.

2.2. TAN mass balance calculations

This section was developed to estimate the required flow to and from a biofilter to maintain a given total ammonia-nitrogen (TAN) concentration in the culture tank and to estimate the amount of water which must be wasted from the system to keep the nitrate-nitrogen at a acceptable level. In cell B19, the spreadsheet estimates the amount of TAN that is generated within the system as a function of the feed input (cell **B14)** and protein content (cell **B18**). The number '0.065' within the equation in cell D19 is a calibrated constant used to approximate a system that yields ammonia nitrogen at a rate of 2.5% of the input feed rate under the given system conditions. In general, approximately 2.0–3.5% of input feed by weight becomes total ammonia-nitrogen (Liao and Mayo, 1974; Wheaton et al., 1994). This section of the spreadsheet can be expanded to include such variables as the digestibility of the feed, a specific species being cultured, or the rate of fecal solids removal. With the TAN production rate determined (displayed in cell B19), the user inputs the desired tank TAN concentration in cell **B21**, and an estimate of the passive nitrification that occurs within the production system in cell **B22.** Passive nitrification refers to ammonia nitrogen that is oxidized to nitrate by bacteria growing on surfaces within the system, other than the biological filter. Passive nitrification can account for as much as 30% of the total nitrification ongoing in a recirculating system. Cell B23 calculates and displays the rate TAN production (after accounting for passive nitrification) that is required to be processed by the biofilter to maintain the user specified tank TAN concentration.

As with passive nitrification, in most recirculation systems, there is usually some amount of passive bacterial denitrification occurring within the system. With high concentrations of nitrate-nitrogen in the system and low dissolved oxygen concentrations in areas where organic solids may build up, a certain volume of the system may become anoxic. In these areas, conditions may be conducive for some amount of bacterial denitrification (Bovendeur et al., 1987). The user inputs an estimate of the amount of passive denitrification ongoing within the system in Cell **B24**. While 10% can be input in most cases, in particularly clean systems or if the user wants to conservatively overestimate the amount of new water needed, zero can be input. In cell **B25** the user inputs the desired maximum nitrate-nitrogen concentration in the tank water.

Intensive recirculating systems are usually defined as those that replace less than 10% of the system volume per day. Unless an active denitrification system is being used, the concentration of nitrate-nitrogen within a recirculating system is usually a function of the amount of waste water discharged and the amount of new water added as makeup. The amount of new water required to maintain the user specified nitrate-nitrogen concentration, (adjusted for passive denitrification) is calculated and displayed in cell B26 of the spreadsheet. This rate of new water addition to the system (equal to the discharge flow rate) will be used to estimate the partial removal of other waste products later in the spreadsheet.

The biological filter (biofilter) removal efficiency input in cell **B28** is generally estimated as a function of numerous operational variables. These include, inflow TAN concentration, temperature, filter surface area, filter type (RBC, fluidized bed, trickling filter), hydraulic loading rate and hydraulic retention time (Losordo and Westers, 1994). The daily flow to the biofilter required to maintain the desired ammonia-nitrogen concentration is calculated in B29 and displayed in cell B30 in liters per minute and B31 in gallons per minute.

2.3. Biofilter sizing calculations

This section of the spreadsheet was developed to help estimate the size of the biofilter that is needed to provide adequate nitrification capacity for the system. The nitrifying capacity of a biofilter is largely determined by the biofilter media that is used and the volume of the biofilter. The surface area of biofilter media is usually measured in terms of m^2 of filter area per m^3 of media and is referred to as the specific surface area of the media. Media with more void space has lower specific surface area. In general, 'aerial' nitrification rates for biofilters used in aquaculture (maximum total ammonia-nitrogen input concentrations of less than 2 mg TAN/l) range from 0.15–1.0 g TAN/m^2/day. In other words, one square meter of filter surface area can oxidize 0.15–1.0 g of total ammonia-nitrogen to nitrate-nitrogen over a 1-day period. Filter medias with low specific surface areas such as trickling filters and rotating biological contactors (specific surface areas of 100–400 m^2/m^3)

often have higher aerial nitrification rates (0.5–1.0 g TAN/m^2/day) (Wheaton et al., 1994). Fluidized bed filters and plastic bead filters have higher specific surface areas, lower void spaces and generally have lower aerial nitrification rates (0.10–0.5 g TAN/m^2/day) (Wheaton et al., 1994; Beecher et al., 1997; Greiner and Timmons, 1998).

In cells **B35** and **B37**, the user inputs the estimated aerial nitrification rate and the specific surface area of the biofilter, respectively. The required surface area is then calculated and displayed in cell B36, with the estimated total volume of media required for the system calculated and displayed in cell B38. The remaining calculations in this section are presented to allow comparison of the cost and dimensions of various biofilter media options.

2.4. Solids mass balance calculations

The user is required to describe three attributes of the system and one desired condition in this section. In cell **B47**, the amount of feed becoming solid waste must be estimated and input. This should include uneaten feed and feces. Depending on the system, feed quality, and species being cultured, this number can range from 25 to 50%. In cell **B49**, the user inputs the concentration of suspended solids that they would like to maintain within the culture tank. The user then inputs the efficiency of the solids removal components used in the recirculating system. If the tank system is equipped with a double drain or internal particle trap, the efficiency of this component to remove waste solids is input in cell **B50**. This efficiency is estimated as a percentage of the total waste solids produced per day. Next, the removal efficiency of a settling basin, bead filter or screen filter component is input in cell **B53**. This efficiency is defined as the percentage of solids removed by the component on a single pass through the device. Finally, the total recirculated flow that is required on a daily basis to keep the suspended solids concentration at the user specified concentration is calculated and displayed in cell B54. This estimate is expressed in liters per minute and gallons per minute in cells B55 and B56, respectively.

2.5. Oxygen mass balance calculations

Dissolved oxygen consumption in recirculating systems is primarily a function of the cultured species, the rate of feed input to the system and the design of the waste treatment system. Dissolved oxygen is consumed through the respiration of both the fish in the system and, to a large extent, the autotrophic (nitrifying bacteria) and heterotrophic bacteria (bacteria responsible for the breakdown of organic matter). How quickly the wasted solids are removed from the system can have a major impact on the amount of oxygen that is consumed. Another key consumption factor is whether the oxygen for nitrification comes from the air (non-submerged filters such as trickling filters or RBC's) or from the water (filters with submerged biofilter media).

The inputs for this section begins with defining whether the biofilter media is submerged in water or not. Entering the number 1 in cell **B61** will alter the equation in cell B67 to calculate the oxygen that is consumed by the nitrifying bacteria of a submerged biological filter. Entering the number 0 will emulate a non-submerged filter where the oxygen for nitrification in the filter comes from the atmosphere.

The next data input is the oxygen consumption of the cultured fish related to feed inputs. Westers (1979) determined that for salmonids, 200–250 g of oxygen are consumed per kg of feed input. Additionally, data presented by Thomas and Piedrahita (1997) indicates that the respiration rate of White Sturgeon (550 g average wt, fed 2.5–3% of their body weight per day) varied from 290–385 g O_2/kg of feed. From these and other studies, we can assume this number varies between 200–500 g O_2/kg of feed (0.2–0.5 kg O_2/kg of feed). This estimated number is input in cell **B62** as a percentage (%).

In cells **B64** and **B65**, the desired concentration of oxygen in the culture tank and in the returning water from the oxygenation system are input, respectively. The concentration of oxygen in the water coming back to the tank from the oxygenation system is a function of the system design and oxygen injection system sizing.

The estimated oxygen utilized within the system including passive nitrification is calculated and displayed in B68. The estimated flow rate required under these conditions is calculated and displayed in cells B69–B71 in the appropriate units.

3. Discussion

This spreadsheet is by no means to be considered a highly refined model that can be used with little or no design experience. While most systems designers may have similar tools in their possession, to date, none have been published. The spreadsheet program is intended to serve as a basis for improved and expanded models for use in the aquaculture industry. In addition to being used as a design tool, this type of spreadsheet can be used to answer some 'what if' type of questions before a system is designed. The designer can estimate the results of changes to the system before time and money is spent. The authors recommend that this design approach be adopted, a prototype system developed, operated and refined, and care then be taken in the scaling up of the system for commercial applications.

References

Beecher, L.E., DeLosReyes, A., Drennan, D.G., Malone, R.F., 1997. Alternative media for enhanced nitrification rates in propeller-washed bead filters. In: Timmons, M.B., Losordo, T.M. (Eds.), Advances in Aquacultural Engineering. Proceedings from the Aquacultural Engineering Society (AES) Technical Sessions at the Fourth International Symposium on Tilapia in Aquaculture, NRAES Publication No. 105. Northeast Regional Agricultural Engineering Service, Riley–Robb Hall, Ithaca, NY, USA, pp. 263–275.

Bovendeur, J., Eding, E.H., Henken, A.M., 1987. Design and performance of a water recirculation system for high-density culture of the African catfish, *clarias gariepinus* (burchell 1822). Aquaculture 63, 329–353.

Greiner, A.D., Timmons, M.B., 1998. Evaluation of the nitrification rates of microbead and trickling filters in an intensive recirculating tilapia production facility. Aquacultural Eng. 18, 189–200.

Hobbs, A., Losordo, T., DeLong, D., Regan, J., Bennett, S., Gron, R., Foster, B., 1997. A commercial, public demonstration of recirculating aquaculture technology: the CP&L/EPRI Fish Barn at North Carolina State University. In: Timmons, M.B., Losordo, T.M. (Eds.), Advances in Aquacultural Engineering. Proceedings from the Aquacultural Engineering Society Technical Sessions at the Fourth International Symposium on Tilapia in Aquaculture. NRAES Publication No. 105. Northeast Regional Agricultural Engineering Service, Riley-Robb Hall, Ithaca, NY, USA, pp. 151–158.

Liao, P.B., Mayo, R.D., 1974. Intensified fish culture combining water reconditioning with pollution abatement. Aquaculture 3, 61–85.

Losordo, T.M., Westers, H., 1994. System carrying capacity. In: Timmons, M.B., Losordo, T.M. (Eds.), Aquaculture Water Reuse Systems: Engineering Design and Management. Developments in Aquaculture and Fisheries Sciences, vol. 27. Elsevier, Amsterdam, pp. 9–60.

Thomas, S.L., Piedrahita, R.H., 1997. Oxygen consumption rates of white sturgeon under commercial culture conditions. Aquaculture Eng. 16, 227–238.

Twarowska, J.G., Westerman, P.W., Losordo, T.M., 1997. Water treatment and waste characterization evaluation in an intensive recirculating fish production system. Aquaculture Eng. 16, 133–147.

Westers, H., 1979. Principles of Intensive Fish Culture. Michigan Department of Natural Resources, Lansing, MI, USA, 108 pp.

Wheaton, F.W., Hochheimer, J.N., Kaiser, G.E., Malone, R.F., Krones, M.J., Libey, G.S., Easters, C.C., 1994. Nitrification filter design methods. In: Timmons, M.B., Losordo, T.M. (Eds.), Aquaculture Water Reuse Systems: Engineering Design and Management. Developments in Aquaculture and Fisheries Sciences, vol. 27. Elsevier, Amsterdam, pp. 127–171.